Iteračné metódy pre systémy lineárnych rovníc

Všeobecná *x*=H*x*+g
Jacobiho metóda
Gauss-Seidlova metóda

ÚLOHA:

Riešte systém rovníc:

$$\begin{array}{rrrrl} -1\,x_1 & +\,4\,x_2 & & -\,1\,x_4 & = 2 \\ +4\,x_1 & -\,1\,x_2 & -\,x_3 & & = 1 \\ -1\,x_1 & & +\,4\,x_3 & -\,1\,x_4 & = 0 \\ & -1\,x_2 & -\,1\,x_3 & +\,4\,x_4 & = 1 \end{array}$$

```
A = {{-1, 4, 0, -1}, {4, -1, -1, 0}, {-1, 0, 4, -1}, {0, -1, -1, 4}};
% // MatrixForm
```

$$\begin{pmatrix} -1 & 4 & 0 & -1 \\ 4 & -1 & -1 & 0 \\ -1 & 0 & 4 & -1 \\ 0 & -1 & -1 & 4 \end{pmatrix}$$

```
n = Length[A]
```

4

```
b = {2, 1, 0, 1}; b // MatrixForm
```

$$\begin{pmatrix} 2 \\ 1 \\ 0 \\ 1 \end{pmatrix}$$

RIEŠENIE PRIAMOU METÓDOU cez LinearSolve :

```
xr = LinearSolve[A, b] // N
% // MatrixForm
```

{0.5, 0.75, 0.25, 0.5}

$$\begin{pmatrix} 0.5 \\ 0.75 \\ 0.25 \\ 0.5 \end{pmatrix}$$

RIEŠENIE POMOCOU ITERAČNEJ SCHÉMY: x = H.x + g

systém rovníc:

$$\begin{array}{rrrrl} -x_1 & +\,4\,x_2 & & -\,x_4 & = 2 & \rightarrow x_1 = & & 4\,x_2 & & -\,x_4 - 2 \\ 4\,x_1 & -\,x_2 & -\,x_3 & & = 1 & \rightarrow x_2 = & 4\,x_1 & & -\,x_3 & -\,1 \\ -x_1 & & +\,4\,x_3 & -\,x_4 & = 0 & \rightarrow x_3 = & x_1 & & -\,3\,x_3 & +\,x_4 + 0 \\ & -x_2 & -\,x_3 & +\,4\,x_4 & = 1 & \rightarrow x_4 = & & +x_2 & +\,x_3 & -\,3\,x_4 + 1 \end{array}$$

Matica H


```
H = {{0, 4, 0, -1}, {4, 0, -1, 0}, {1, 0, -3, 1}, {0, 1, 1, -3}}; H // MatrixForm
```

$$\begin{pmatrix} 0 & 4 & 0 & -1 \\ 4 & 0 & -1 & 0 \\ 1 & 0 & -3 & 1 \\ 0 & 1 & 1 & -3 \end{pmatrix}$$

Vektor g

```
g = {-2, -1, 0, 1}; g // MatrixForm
```

$$\begin{pmatrix} -2 \\ -1 \\ 0 \\ 1 \end{pmatrix}$$

Štartovací vektor x(0)   a   iteračná schéma:

```
Clear[x]
x[0] = {0, 0, 0, 0}
pit = 20
Do[
 x[k + 1] = H .x[k] + g // N;
 Print[k + 1, ". iteracia: x=", x[k + 1]], {k, 0, pit}]
```

{0, 0, 0, 0}

20

1. iteracia: x={-2., -1., 0., 1.}

2. iteracia: x={-7., -9., -1., -3.}

3. iteracia: x={-35., -28., -7., 0.}

4. iteracia: x={-114., -134., -14., -34.}

5. iteracia: x={-504., -443., -106., -45.}

6. iteracia: x={-1729., -1911., -231., -413.}

7. iteracia: x={-7233., -6686., -1449., -902.}

8. iteracia: x={-25 844., -27 484., -3788., -5428.}

9. iteracia: x={-104 510., -99 589., -19 908., -14 987.}

10. iteracia: x={-383 371., -398 133., -59 773., -74 535.}

11. iteracia: x=$\{-1.518\times10^{6}, -1.47371\times10^{6}, -278587., -234300.\}$

12. iteracia: x=$\{-5.66055\times10^{6}, -5.79341\times10^{6}, -916538., -1.0494\times10^{6}\}$

13. iteracia: x=$\{-2.21242\times10^{7}, -2.17257\times10^{7}, -3.96033\times10^{6}, -3.56175\times10^{6}\}$

14. iteracia: x=$\{-8.33409\times10^{7}, -8.45366\times10^{7}, -1.3805\times10^{7}, -1.50007\times10^{7}\}$

15. iteracia: x=$\{-3.23146\times10^{8}, -3.19559\times10^{8}, -5.69267\times10^{7}, -5.33394\times10^{7}\}$

16. iteracia: x=$\{-1.2249\times10^{9}, -1.23566\times10^{9}, -2.05705\times10^{8}, -2.16467\times10^{8}\}$

17. iteracia: x=$\{-4.72616\times10^{9}, -4.69387\times10^{9}, -8.24246\times10^{8}, -7.91961\times10^{8}\}$

18. iteracia: x=$\{-1.79835\times10^{10}, -1.80804\times10^{10}, -3.04538\times10^{9}, -3.14224\times10^{9}\}$

19. iteracia: x=$\{-6.91793\times10^{10}, -6.88888\times10^{10}, -1.19896\times10^{10}, -1.16991\times10^{10}\}$

20. iteracia: x=$\{-2.63856\times10^{11}, -2.64728\times10^{11}, -4.49095\times10^{10}, -4.57812\times10^{10}\}$

21. iteracia: x=$\{-1.01313\times10^{12}, -1.01051\times10^{12}, -1.74909\times10^{11}, -1.72294\times10^{11}\}$

Iteračná schéma diverguje

Veta. Iteračná schéma $x^{(k+1)} = H\,x^{(k)} + g$ konverguje práve vtedy, keď $\rho(H)<1$. $\rho(H)=\max\{\,|\lambda_i|\,\},\ i = 1, \ldots n$

```
Eigenvalues[H]
```

$\{-5, 1+2\sqrt{2}, -3, 1-2\sqrt{2}\}$

```
Eigenvalues[H] // N
```

{-5., 3.82843, -3., -1.82843}


```
ro = Max[Abs[Eigenvalues[H]]]
```

5

Iteračná schéma diverguje, ρ>1

Jacobiho metóda a Gauss - Seidlova metóda

Vlastnosť I.

Ak matica A je diagonálne ostro dominantná ($|a_{i\,i}| > \sum_{j=1}^{n} a_{i\,j},\ i \neq j$), Jacobiho aj Gauss - Seidlova metóda konvergujú.

Tieto metódy niekedy konvergujú aj keď táto podmienka nie je splnená. Je však nutné, aby prvky na hlavnej diagonále boli v absolútnej hodnote väčšie ako ostatné prvky matice.

Vlastnosť II.

Ak matica A je symetrická ($A = A^T$) and positivne definitná ($x^T.A.x > 0,\ x \neq \bar{0}$, Jacobiho aj Gauss - Seidlova metóda konvergujú.

Pôvodná matica A nie je diagonálne dominantná ani symetrická. Avšak ak vymeníme poradie prvých dvoch rovníc, odpovedajúce riadky v matici A a vo vektore b sa vymenia, Zmenená matica systému už diagonálne dominantná je. Naviac je aj symetrická.

$$A = \begin{pmatrix} -1 & 4 & 0 & -1 \\ 4 & -1 & -1 & 0 \\ -1 & 0 & 4 & -1 \\ 0 & -1 & -1 & 4 \end{pmatrix} = \begin{pmatrix} 4 & -1 & -1 & 0 \\ -1 & 4 & 0 & -1 \\ -1 & 0 & 4 & -1 \\ 0 & -1 & -1 & 4 \end{pmatrix}$$

Matica A = M+Diag+NH

```
A = {{4, -1, -1, 0}, {-1, 4, 0, -1}, {-1, 0, 4, -1}, {0, -1, -1, 4}};
% // MatrixForm
b = {1, 2, 0, 1};
% // MatrixForm
```

$$\begin{pmatrix} 4 & -1 & -1 & 0 \\ -1 & 4 & 0 & -1 \\ -1 & 0 & 4 & -1 \\ 0 & -1 & -1 & 4 \end{pmatrix}$$

$$\begin{pmatrix} 1 \\ 2 \\ 0 \\ 1 \end{pmatrix}$$

```
Diag = DiagonalMatrix[{4, 4, 4, 4}]; Diag // MatrixForm
```

$$\begin{pmatrix} 4 & 0 & 0 & 0 \\ 0 & 4 & 0 & 0 \\ 0 & 0 & 4 & 0 \\ 0 & 0 & 0 & 4 \end{pmatrix}$$

```
M = {{0, 0, 0, 0}, {-1, 0, 0, 0}, {-1, 0, 0, 0}, {0, -1, -1, 0}}; M // MatrixForm
```

$$\begin{pmatrix} 0 & 0 & 0 & 0 \\ -1 & 0 & 0 & 0 \\ -1 & 0 & 0 & 0 \\ 0 & -1 & -1 & 0 \end{pmatrix}$$

```
NH = {{0, -1, -1, 0}, {0, 0, 0, -1}, {0, 0, 0, -1}, {0, 0, 0, 0}}; NH // MatrixForm
```

$$\begin{pmatrix} 0 & -1 & -1 & 0 \\ 0 & 0 & 0 & -1 \\ 0 & 0 & 0 & -1 \\ 0 & 0 & 0 & 0 \end{pmatrix}$$

```
A == M + Diag + NH
```

True

Jacobiho metóda
Cyklus cez matice

$$x^{k+1} = -D^{-1}.(M+N).x^k + D^{-1}.b$$
$$H = -D^{-1}.(M+N), \;\; g = D^{-1}.b$$

```
H = -Inverse[Diag].(M + NH); H // MatrixForm
```

$$\begin{pmatrix} 0 & \frac{1}{4} & \frac{1}{4} & 0 \\ \frac{1}{4} & 0 & 0 & \frac{1}{4} \\ \frac{1}{4} & 0 & 0 & \frac{1}{4} \\ 0 & \frac{1}{4} & \frac{1}{4} & 0 \end{pmatrix}$$

```
Eigenvalues[H]
```

$\left\{-\frac{1}{2}, \frac{1}{2}, 0, 0\right\}$

The Jacobiho metóda bude konvergovať, ρ<1, ρ = 1/2.

```
g = Inverse[Diag].b
```

$\left\{\frac{1}{4}, \frac{1}{2}, 0, \frac{1}{4}\right\}$


```
Clear[x]
x[0] = {0, 0, 0, 0};
it = 20;
Do[
 x[k + 1] = H.x[k] + g;
 Print[k + 1, ". iterácia: x= ", x[k + 1] // N], {k, 0, it}]
```

1. iterácia: x= {0.25, 0.5, 0., 0.25}

2. iterácia: x= {0.375, 0.625, 0.125, 0.375}

3. iterácia: x= {0.4375, 0.6875, 0.1875, 0.4375}

4. iterácia: x= {0.46875, 0.71875, 0.21875, 0.46875}

5. iterácia: x= {0.484375, 0.734375, 0.234375, 0.484375}

6. iterácia: x= {0.492188, 0.742188, 0.242188, 0.492188}

7. iterácia: x= {0.496094, 0.746094, 0.246094, 0.496094}

8. iterácia: x= {0.498047, 0.748047, 0.248047, 0.498047}

9. iterácia: x= {0.499023, 0.749023, 0.249023, 0.499023}

10. iterácia: x= {0.499512, 0.749512, 0.249512, 0.499512}

11. iterácia: x= {0.499756, 0.749756, 0.249756, 0.499756}

12. iterácia: x= {0.499878, 0.749878, 0.249878, 0.499878}

13. iterácia: x= {0.499939, 0.749939, 0.249939, 0.499939}

14. iterácia: x= {0.499969, 0.749969, 0.249969, 0.499969}

15. iterácia: x= {0.499985, 0.749985, 0.249985, 0.499985}

16. iterácia: x= {0.499992, 0.749992, 0.249992, 0.499992}

17. iterácia: x= {0.499996, 0.749996, 0.249996, 0.499996}

18. iterácia: x= {0.499998, 0.749998, 0.249998, 0.499998}

19. iterácia: x= {0.499999, 0.749999, 0.249999, 0.499999}

20. iterácia: x= {0.5, 0.75, 0.25, 0.5}

21. iterácia: x= {0.5, 0.75, 0.25, 0.5}

Jacobiho schéma konverguje a riešenie dosiahne v 20-tej iterácii.

Nájdite reziduum A.x - b pre 10-tu iteráciu x, and jej presnosť (vzdialenosť medzi x(10) a x(9)).

```
r = A.x[10] - b
```

$\left\{-\frac{1}{1024}, -\frac{1}{1024}, -\frac{1}{1024}, -\frac{1}{1024}\right\}$

```
nrJ = Sqrt[r.r]
```

$\frac{1}{512}$

```
nrJ // N
```

0.00195313

```
d = x[10] - x[9]
```

$\left\{\frac{1}{2048}, \frac{1}{2048}, \frac{1}{2048}, \frac{1}{2048}\right\}$

```
ndJ = Sqrt[d.d]
```

$\frac{1}{1024}$

```
ndJ // N
```

0.000488281

Pre Jacobiho metódu norma (veľkosť) rezidua 10 - tej iterácie vektora *x* je 0.00195313 a presnosť 0.000488281 .

Gauss-Seidel method

Cyklus cez matice $x^{k+1} = -(M+D)^{-1}.N.x^k + (M+D)^{-1}.b$

$$H = -(M+D)^{-1}.N, \; g = (M+D)^{-1}.b$$

```
H = -Inverse[M + Diag].NH
```

$\left\{\left\{0, \frac{1}{4}, \frac{1}{4}, 0\right\}, \left\{0, \frac{1}{16}, \frac{1}{16}, \frac{1}{4}\right\}, \left\{0, \frac{1}{16}, \frac{1}{16}, \frac{1}{4}\right\}, \left\{0, \frac{1}{32}, \frac{1}{32}, \frac{1}{8}\right\}\right\}$

```
Eigenvalues[H]
```

$\left\{\frac{1}{4}, 0, 0, 0\right\}$

Gauss -Seidlova metóda konverguje, ρ<1, ρ = 1/4.

```
g = Inverse[M + Diag].b
```

$\left\{\frac{1}{4}, \frac{9}{16}, \frac{1}{16}, \frac{13}{32}\right\}$


```
Clear[x]
x[0] = {0, 0, 0, 0};
it = 20;
Do[
 x[k + 1] = H .x[k] + g;
 Print[k + 1, ". iterácia: x=", x[k + 1] // N], {k, 0, it}]
```

1. iterácia: x={0.25, 0.5625, 0.0625, 0.40625}

2. iterácia: x={0.40625, 0.703125, 0.203125, 0.476563}

3. iterácia: x={0.476563, 0.738281, 0.238281, 0.494141}

4. iterácia: x={0.494141, 0.74707, 0.24707, 0.498535}

5. iterácia: x={0.498535, 0.749268, 0.249268, 0.499634}

6. iterácia: x={0.499634, 0.749817, 0.249817, 0.499908}

7. iterácia: x={0.499908, 0.749954, 0.249954, 0.499977}

8. iterácia: x={0.499977, 0.749989, 0.249989, 0.499994}

9. iterácia: x={0.499994, 0.749997, 0.249997, 0.499999}

10. iterácia: x={0.499999, 0.749999, 0.249999, 0.5}

11. iterácia: x={0.5, 0.75, 0.25, 0.5}

12. iterácia: x={0.5, 0.75, 0.25, 0.5}

13. iterácia: x={0.5, 0.75, 0.25, 0.5}

14. iterácia: x={0.5, 0.75, 0.25, 0.5}

15. iterácia: x={0.5, 0.75, 0.25, 0.5}

16. iterácia: x={0.5, 0.75, 0.25, 0.5}

17. iterácia: x={0.5, 0.75, 0.25, 0.5}

18. iterácia: x={0.5, 0.75, 0.25, 0.5}

19. iterácia: x={0.5, 0.75, 0.25, 0.5}

20. iterácia: x={0.5, 0.75, 0.25, 0.5}

21. iterácia: x={0.5, 0.75, 0.25, 0.5}

Gauss -Seidlová schéma konverguje a riešenie dosiahne v 11-tej iterácii.

Nájdite reziduum A.x - b pre 10-tu iteráciu x, and jej presnosť (vzdialenosť medzi x(10) a x(9)).

```
r = A.x[10] - b
```

$\left\{-\frac{9}{2097152}, -\frac{9}{8388608}, -\frac{9}{8388608}, 0\right\}$

```
nrGS = Sqrt[r.r]
```

$\frac{27}{4194304\sqrt{2}}$

```
nrGS // N
```

4.55186×10^{-6}

```
d = x[10] - x[9]
```

$\left\{\frac{9}{2097152}, \frac{9}{4194304}, \frac{9}{4194304}, \frac{9}{8388608}\right\}$

```
ndGS = Sqrt[d.d]
```

$\frac{45}{8\,388\,608}$

```
ndGS // N
```

5.36442×10^{-6}

Pre Gauss - Seidlovu metódu norma (veľkosť) rezidua 10 - tej iterácie vektora x je 4.55186×10^{-6} a presnosť 5.36442×10^{-6} .

Porovnajte vypočítané veľkosti rezuíd A.x - b 10-tej iterácie pre obidve metódy.

```
nrGS // N
```

4.55186×10^{-6}

```
nrJ // N
```

0.00195313

Veľkosť rezidua v Gauss -Seidlovej metóde 4.55186×10^{-6} je oveľa menšia ako veľkosť rezidua pre Jacobiho metódu 0.00195313.

▪ Cyklus s výstupnou podmienkou: norma $||x^{k+1}\text{-}x^k|| < \delta$

```
delta = 10^-4;
Clear[x]
x[0] = {0, 0, 0, 0};
it = 20;
Do[
 x[k + 1] = H .x[k] + g;
 Print[k + 1, ". iterácia: x=", x[k + 1] // N];
 d = x[k + 1] - x[k];
 If[Sqrt[d.d] < delta, Break[]],
 {k, 0, it}]
Print["norma ||x^(k+1)-x^k||= ", Sqrt[d.d] // N]
```

```
1. iterácia: x={0.25, 0.5625, 0.0625, 0.40625}
2. iterácia: x={0.40625, 0.703125, 0.203125, 0.476563}
3. iterácia: x={0.476563, 0.738281, 0.238281, 0.494141}
4. iterácia: x={0.494141, 0.74707, 0.24707, 0.498535}
5. iterácia: x={0.498535, 0.749268, 0.249268, 0.499634}
6. iterácia: x={0.499634, 0.749817, 0.249817, 0.499908}
7. iterácia: x={0.499908, 0.749954, 0.249954, 0.499977}
8. iterácia: x={0.499977, 0.749989, 0.249989, 0.499994}
norma ||x^(k+1)-x^k||= 0.0000858307
```


▪ Cyklus s výstupnou podmienkou: norma rezidua $||A\,.x^{k+1}\text{-}b|| < \delta$

```
delta = 10^-4;
Clear[x]
x[0] = {0, 0, 0, 0};
it = 20;
Do[
 x[k + 1] = H .x[k] + g;
 Print[k + 1, ". iterácia: x=", x[k + 1] // N];
 res = A.x[k + 1] - b;
 If[Sqrt[res.res] < delta, Break[]],
 {k, 0, it}]
Print["norma rezidua ||A .x^(k+1)-b||= ", Sqrt[res.res] // N]
```

```
1. iterácia: x={0.25, 0.5625, 0.0625, 0.40625}
2. iterácia: x={0.40625, 0.703125, 0.203125, 0.476563}
3. iterácia: x={0.476563, 0.738281, 0.238281, 0.494141}
4. iterácia: x={0.494141, 0.74707, 0.24707, 0.498535}
5. iterácia: x={0.498535, 0.749268, 0.249268, 0.499634}
6. iterácia: x={0.499634, 0.749817, 0.249817, 0.499908}
7. iterácia: x={0.499908, 0.749954, 0.249954, 0.499977}
8. iterácia: x={0.499977, 0.749989, 0.249989, 0.499994}
norma rezidua ||A .x^(k+1)-b||= 0.0000728298
```

ÚLOHA: Vyskúšajte iný štartovací vektor

▪ Cyklus s výstupnou podmienkou: norma rezidua $||A\,.x^{k+1}\text{-}b|| < \delta$

```
delta = 10^-4;
Clear[x]
x[0] = {1, 2, 3, -4};
it = 20;
Do[
 x[k + 1] = H .x[k] + g;
 Print[k + 1, ". iterácia: x=", x[k + 1] // N];
 res = A.x[k + 1] - b;
 If[Sqrt[res.res] < delta, Break[]],
 {k, 0, it}]
Print["norma rezidua ||A .x^(k+1)-b||= ", Sqrt[res.res] // N]
```

```
1. iterácia: x={1.5, -0.125, -0.625, 0.0625}
2. iterácia: x={0.0625, 0.53125, 0.03125, 0.390625}
3. iterácia: x={0.390625, 0.695313, 0.195313, 0.472656}
4. iterácia: x={0.472656, 0.736328, 0.236328, 0.493164}
5. iterácia: x={0.493164, 0.746582, 0.246582, 0.498291}
6. iterácia: x={0.498291, 0.749146, 0.249146, 0.499573}
7. iterácia: x={0.499573, 0.749786, 0.249786, 0.499893}
8. iterácia: x={0.499893, 0.749947, 0.249947, 0.499973}
9. iterácia: x={0.499973, 0.749987, 0.249987, 0.499993}
norma rezidua ||A .x^(k+1)-b||= 0.000084968
```